
1 About the XNS Public Trust Organization (XNSORG) 78 

XNSORG is a non-profit organization created to develop and manage Extensible Name Service 79 
(XNS) in the public interest. XNS is an open, XML-based addressing system and data exchange 80 
protocol for identifying and linking any resource participating in any kind of digital transaction. The 81 
complete technical specificatons for XNS are available at the XNSORG website at 82 
http://www.xns.org.  83 

XNSORG has been granted intellectual property rights by contributors to the XNS specifications, 84 
and in turn makes these rights available to the public under an open, royalty-free license as described 85 
in http://www.xns.org/pages/XNS_License.pdf.  86 

2 About this Document 87 

This document is the formal specification for XNS Addressing Specification v1.1. It is a minor 88 
update to the XNS Addressing Specification portion of the XNS Technical Specifications v1.0, 89 
available in their entirety at http://www.xns.org/pages/XNS_Technical_Specs.pdf.  90 

Since the abstract addressing concepts in XNS are useful outside of the scope of XNS itself, this 91 
specification is being published as a standalone document so it can be referenced independently from 92 
the full XNS specifications. 93 

3 Terminology and Conventions 94 

The following conventions are used in this document: 95 

• The keywords "MUST", "MUST NOT", "REQUIRED", "SHALL", "SHALL NOT", 96 
"SHOULD", "SHOULD NOT", "RECOMMENDED", "MAY", and "OPTIONAL" as used 97 
in this document are to be interpreted as described in IETF RFC-2119 [1].  98 

• EBNF productions use the EBNF syntax notation as described in the W3C XML 1.0 99 
Recommendation [2]. 100 

4 Introduction to XNS Addressing 101 

The W3C XPath 1.0 Recommendation [3] establishes a standard syntax for addressing the nodes of a 102 
structured XML document. Since XNS Addressing provides addresses for a network of linked XML 103 
documents, it has a similar need for a standardized syntax. 104 

However unlike XPath, which was designed primarily for programmatic use (and includes many 105 
additional functions for querying data sets within an XML document), XNS addressing must fulfill 106 
requirements for machine efficiency, human usability, and identity persistence. For a complete 107 
discussion of those requirements, please see the XNSORG white paper From Name Service to 108 
Identity Service: How XNS builds on the DNS Model.  109 

This specification provides the normative rules for XNS address validity. It includes four rule sets: 110 

1.  The EBNF definition of XNS addressing syntax. 111 

2.  The EBNF definition for URI encoding of XNS addresses and XNS service invocations.  112 

3.  XNS ID normalization rules. 113 

4.  XNS Name normalization rules. 114 

5 EBNF Definition of XNS Addressing Syntax 115 

Following is the authoritative EBNF definition of an XNS address (see the Notation section of the 116 
W3C XML 1.0 Recommendation [2] for a summary of the EBNF syntax). All XNS addressing 117 
values used in XNS implementations MUST conform to this EBNF definition. 118 

[ 1] XNSAddress ::= XNSID | XNSName | AbsoluteAddress 119 
[ 2] XNSID ::= IdentityID | IdentityDataID | DataID | RelativeDataID 120 
[ 3] IdentityID = ':' [HostIDNode *('.' HostIDNode)] ':' [IdentityIDNode *('.' 121 
IdentityIDNode)] 122 
[ 4] HostIDNode ::= INT | ('(' IdentityID | URN ')')  123 
[ 5] INT ::= Non-negative integer 124 
[ 6] URN ::= Uniform Resource Name as specified in IETF RFC 2141 [5]   125 
[ 7] IdentityIDNode ::= ID | ('(' IdentityID | URN ')') 126 
[ 8] ID ::= XML character string normalized according to the XNS ID Normalization 127 
Rules 128 
[ 9] IdentityDataID ::= IdentityID DataID 129 
[10] DataID ::= ';' DataIDNode [RelativeDataID] 130 
[11] DataIDNode ::= ID | ('(' (IdentityDataID  | URN) ')')  131 
[12] RelativeDataID ::= *('.' DataIDNode) [',' Version] 132 
[13] Version ::= ('v' VersionNumber) | ('t' VersionDate) 133 
[14] VersionNumber ::= Non-negative integer 134 
[15] VersionDate ::= XML DateTime instance 135 
[16] XNSName ::= IdentityName | IdentityDataName | DataName | RelativeDataName 136 
[17] IdentityName ::= (NamespaceSymbol | '//') IdentityNameNode *('/' 137 
IdentityNameNode) 138 
[18] NamespaceSymbol ::= '=' | '@' | '+' 139 
[19] IdentityNameNode ::= Name | ('(' IdentityAddress | URI ')') 140 
[20] Name ::= XML character string normalized according to the XNS Name 141 
Normalization Rules 142 
[21] IdentityAddress ::= (IdentityID ['!' IdentityName]) | IdentityName 143 
[22] URI ::= Uniform Resource Identifier as specified in IETF RFC 2396 [4] 144 
[23] IdentityDataName ::= IdentityName DataName 145 
[24] DataName ::= '/' RelativeDataName 146 
[25] RelativeDataName ::= DataNameNode *('/' DataNameNode) ['/,' Version] 147 
[26] DataNameNode  ::= Name | ('(' (IdentityDataAddress | URI) ')') 148 
[27] IdentityDataAddress ::= (IdentityDataID ['!' IdentityDataName]) | 149 
IdentityDataName 150 
[28] AbsoluteAddress ::= IdentityAddress | IdentityDataAddress 151 

5.1 Key Concepts in the EBNF 152 

The EBNF is based on a handful of concepts that are repeated throughout the productions. The 153 
following sections explain these key concepts prior to the line-by-line documentation. 154 

5.1.1 IDs, Names and Addresses 155 

Three of the most fundamental requirements of XNS addressing are the ability to:  156 

1. Provide an abstraction layer capable of representing the identity of any network actor or 157 
entity—machine, network location, application, user, business, taxonomy category, etc.,  158 

2. Enable this identity to persist for the lifetime of the resource it represents, and 159 

3. Enable this identity abstraction layer to be federated across any number of communities for 160 
fully decentralized, delegated identity management. 161 

To meet these requirements XNS addressing follows the architectural principle of semantic 162 
abstraction—separating non-persistent semantic identifiers (names) from persistent abstract 163 

identifiers (IDs). In most computer naming systems, a name is resolved directly to the physical 164 
location of a resource—a file on a disk, a host machine on a network, a record in a database. In XNS 165 
addressing a name is normally resolved to an XNS ID, which in turn resolves to the network location 166 
of the identity document or a node within it. This network location is expressed as a Uniform 167 
Resource Identifier (URI) [4]. Since URIs do not require persistence of the address, an XNS ID 168 
meets the higher persistence requirements of a Uniform Resource Name (URN) [5]. 169 

Since the address of an identity may use a name, an ID, or both, XNS addressing supports all three 170 
concepts: 171 

• IDs are persistent addressing values intended primarily for machine use. XNS IDs are 172 
permanent identifiers that can be either local or global in scope, but which never change once 173 
they are assigned to an XNS identity or an identity attribute. An ID is a URN, i.e., it may 174 
expire, but it may never be assigned to another identity or identity attribute. Likewise, if the 175 
identity or identity attribute is deleted from the system, the ID or IDs used to identify it are 176 
retired and never reused. XNS IDs are the basis for all persistent relationships in XNS, 177 
whether references or links. 178 

• Names are non-persistent addressing values intended primarily for human use. XNS names 179 
typically represent semantic relationships that can change as real-world identity names and 180 
relationships change, so they do not have the same persistence requirements as XNS IDs. 181 
XNS naming is implemented as an abstraction layer on top of XNS IDs, i.e., an XNS name 182 
usually resolves to an XNS ID before it is resolved to the network location of the target, e.g., 183 
a URI.  184 

• Addresses are a composite addressing type that can consist of either an XNS ID or an XNS 185 
name or a combination of both. In the latter case the XNS ID is authoritative and the XNS 186 
name always serves as a human-readable comment. 187 

5.1.2 Object Versions 188 

Maintaining state is necessary to support the requirements of being able to uniquely address, share, 189 
and synchronize identity attribute values. Identity documents must be able to unambiguously identify 190 
different versions of identity attributes at different moments in time. To unambiguously address a 191 
specific version of an identity attribute, the EBNF "Version" productions allow the version value 192 
identifying the target version to be appended to the XNS ID or XNS name.  193 

This solves a longstanding problem with URI syntax: how to maintain the persistent identity of a 194 
resource while still being able to authoritatively specify a particular version of that resource. Under 195 
current W3C specifications such as P3P, a different URI must be used to specify a different version 196 
of a resource such as a privacy policy. This is necessary because URI syntax does not specify a 197 
versioning component, so the portion of a URI that must change to reflect version changes can only 198 
be established by local convention. 199 

XNS addressing syntax solves this problem by providing an explicit global versioning component. 200 
This version value can be in one of two formats:  201 

• Version Number. This is an integer representing the version of the identity attribute. Note 202 
that there is no requirement that version numbers be sequential; simply that they increase in 203 
value. This allows numeric equivalents of the version date to be used as version numbers. 204 

• Version Date. This is a dateTime instance (as specified by W3C XML Schemas Part 2 [6]). 205 
See the Version Date Format rule for more about this format. 206 

5.1.3 Identities and Identity Data 207 

Absolute and relative are concepts that apply to almost any addressing system. Absolute addresses 208 
are globally unique and can always be resolved regardless of the current addressing context (i.e., they 209 
have a known starting point). By contrast, relative addresses are not globally unique and can only be 210 
resolved relative to the current addressing context. In XNS addressing the concepts of absolute and 211 
relative are modeled by the concepts of identity and data. An XNS identity is always an absolute 212 
identity, capable of being globally independent of any other identity, while any data contained by 213 
that identity—the set of attributes of the identity or its relationship to other identities—are relative to 214 
the identity, since they do not logically make sense outside of that context. 215 

From an addressing perspective this means an XNS identity is conceptually similar to a disk drive in 216 
a file system or a network drive in Unix, while the data contained by an identity is conceptually 217 
similar to the files contained on this drive. XNS simply abstracts the concept of “drive” to any 218 
identifiable container of data—the identity document is the abstract representation of that top- level 219 
container. All nodes below the root node of the identity document represent the attributes (data) of 220 
this container. 221 

Since all XNS identities are absolute, they require absolute addresses, and registering and resolving 222 
these addresses may require inter-identity communications because the addresses may span multiple 223 
identity documents. By contrast, the address of any data within an identity document is always 224 
relative to the root node of that identity document and can therefore be resolved entirely by the 225 
authoritative XNS identity, the same way locating a filename on a local disk drive does not require 226 
any calls to the outside network. 227 

To represent these concepts, the following four terms are used consistently throughout the EBNF: 228 

• Identity is used as the prefix for all absolute values—globally unique IDs, names, or 229 
addresses—that resolve to the root node of an identity document. 230 

• IdentityData is used as the prefix for all absolute values—globally unique IDs, names, or 231 
addresses— that resolve to any lower-level node within an identity document. 232 

• Data is used as the prefix for IDs or names that are not globally unique, but unique only 233 
relative to the root node of an identity document.  234 

• RelativeData is used as the prefix for IDs or names that are unique only relative to the 235 
current node of an identity document. 236 

5.1.4 Host Identities and Hosted Identities 237 

An XNS identity can represent any identifiable entity, from a person to a taxonomy category. 238 
However because an XNS identity may be represented by a resource that physically resides 239 
somewhere on the network, an XNS identity address, if resolvable, must resolve to the network 240 
address of this resource. 241 

In XNS, a special type of identity called a host identity represents the identity of a network 242 
endpoint—a device with a physical network address at which an identity may be contacted. A host 243 
identity is simply an XNS identity with at least one known set of attributes: a list of URIs over which 244 
this host machine accepts messages. 245 

A host identity can be standalone (self-hosted), or it can host any number of other XNS identities 246 
called hosted identities. The collection of the host identity and all hosted identities is called a host 247 
community. Every identity in a host community includes the host identity's address in its own XNS 248 

address just like every web page in a web site includes the same base DNS address (e.g., 249 
“www.example.com/”).  250 

It is important to point out that a host identity may not be associated with any identity it hosts any 251 
more than the operator of a web server is associated with the identity of any web site that runs on it. 252 
While a host identity has its own XNS address, and can store and manage any of the attributes that 253 
profile the host device or operating environment (including its trust credentials), it may not have 254 
anything else in common with the identities it hosts besides being co-located at the same network 255 
endpoint. 256 

5.1.5 Cross-references 257 

 258 

The final key concept in XNS addressing architecture is one that is critical to a distributed, federated 259 
identity system. It is also one of the most novel aspects of XNS addressing. There are times it is 260 
necessary to refer to an identity in a particualar namespace by the address by which that identity is 261 
known in a different namespace. For example, imagine that Alice has a corporate email address of 262 
alice@abc.com. Alice changes jobs and gets a new email address of alice.smith@newco.com. Bob 263 
knows Alice's old address and he knows she now works at newco.com, but he doesn't know her new 264 
address. Alice's new company, newco.com, knows Alice's old address. It would be nice if Bob could 265 
somehow send mail to "that identity at newco.com that is known as alice at abc.com." In other 266 
words, Bob would like to address an identity in the newco.com namespace in terms of that identity's 267 
address in the abc.com namespace. XNS addressing provides this type of cross-community 268 
addressing via a feature called cross-references. 269 

Cross-references are supported syntically by enclosing them in parentheses. The value inside the 270 
parentheses is either a fully qualified XNS Address of a fully qualified URI or URN. For example, 271 
the mailto URI scheme does not support cross-references, but if it did, a mailto address that 272 
incorporated the cross-reference described above might look something like  273 

mailto:(mailto:alice@abc.com)@newco.com 274 
 275 

Obviously this would only work if newco.com could make sense of the cross-reference. That is, 276 
newco.com would need some way to map the cross-reference (mailto:alice@a.com) to alice_smith, 277 
Alice's name in the newco.com namespace. 278 

Broadly then, XNS addressing provides the ability to express logical equivalence so the same 279 
identity or identity attribute can be recognized across multiple host communities. (Note that this 280 
behavior is not required, and in fact may be expressly prohibited when an identity controller wishes 281 
to remain anonymous or pseudonymous). In XNS any element of an identity document, including the 282 
document root object representing the identity itself, can be cross-referenced with another logically 283 
equivalent element in a different identity document. Furthermore, XNS addressing also allows any 284 
URN to be used as a cross-reference for an XNS ID, and any URI to be used as a cross-reference for 285 
an XNS name. 286 

To support the ability to make cross-references, the EBNF productions include the following special 287 
terms for the syntax elements where cross-references can be used: 288 

• IdentityIDNode is the term used for any node in an XNS ID path that terminates in the root 289 
node of an identity document. IdentityIDNodes can be addressed by either a local ID, a cross-290 
reference ID, or a URN. 291 

• DataIDNode is the term used for any node in an XNS ID path that terminates in a node 292 
below the root node of an identity document. DataIDNodes can be addressed by either a local 293 
ID, a cross-reference ID, or a URN. 294 

• IdentityNameNode is the term used for a node in an XNS name path that terminates in the 295 
root node of an identity document. IdentityNameNodes can be addressed by either a local 296 
name, a cross-referenced XNS address (either an XNS ID or XNS name) or a URI. 297 

• DataNameNode is the term used for a node in an XNS ID that terminates in any node below 298 
the root node of an identity document. IdentityNameNodes can be addressed by either a local 299 
name, a cross-referenced XNS address (either an XNS ID or XNS name) or a URI. 300 

5.2 Line-By-Line Documentation of the EBNF 301 

Using the key concepts explained above, the following sections step through the EBNF productions 302 
to explain the structure of an XNS address in detail. 303 

5.2.1 XNS Addresses 304 

[ 1] XNSAddress ::= XNSID | XNSName | AbsoluteAddress 305 
 306 
An XNS address can be one of three overall types. The first two, XNSID and XNSName, are atomic. 307 
The third, AbsoluteAddress, is a composite of an XNS ID value or an XNS name value (or both) that 308 
forms the absolute address of an XNS identity document or a data node within it. 309 

5.2.2 XNS IDs 310 

[ 2] XNSID ::= IdentityID | IdentityDataID | DataID | RelativeDataID 311 
 312 
As explained in the IDs, Names, and Addresses section above, an XNS ID is a permanent semantic 313 
identifier of any identity or identity attribute. It can be one of four types depending on whether it is 314 
the absolute ID for the root node of an identity document (IdentityID), the absolute ID for an element 315 
below the root node of an identity document (IdentityDataID), an ID relative to the root node of an 316 
identity document (DataID), or an ID relative to the current node of an identity document (Rel-317 
ativeDataID). Each of these four types is explained in the following sections. 318 

5.2.3 Identity IDs 319 

[ 3] IdentityID = ':' [ HostIDNode *('.' HostIDNode)] ':' [IdentityIDNode *('.' 320 
IdentityIDNode)] 321 

[ 4] HostIDNode ::= INT | ('(' IdentityID | URN ')') 322 
[ 5] INT ::= Non-negative integer 323 
[ 6] URN ::= Uniform Resource Name as specified in IETF RFC 2141 [3] 324 
[ 7] IdentityIDNode ::= ID | ('(' IdentityID | URN ')') 325 
[ 8] ID ::= XML character string normalized according to the XNS ID Normalization 326 
Rules 327 
 328 
An IdentityID is a fully formed address to an identity (or to the root node of the abstract document 329 
representing an identity). An IdentityID consists of a path that begins with a colon representing the 330 
abstract XNS ID community identity. This is followed by zero or more dot separated HostIDNodes 331 
that, taken together, form a globally unique value representing the host identity. Line 4 defines a 332 
HostIDNode as either 1) an integer or 2) a cross-reference containing an IdentityID or a URN. Note 333 
that it is possible to have no HostIDNodes, in which case the host is the XNS ID community identity.  334 

If  the HostIDNodes in an IdentityID are all integers (i.e. the host identity is expressed without cross-335 
references), the host identity is in the form of an Object ID (OID). An OID is a dot-delimited path of 336 
non-negative integer values that are commonly used in directory systems such as X.500 and LDAP. 337 
Each integer in an OID path must be unique relative to its parent node. In the case of XNS, the OID 338 
root is the XNS ID community identity managed by XNSORG on behalf of the global community. 339 

If a host is identified by a different addressing community, it must have a cross-reference for its first-340 
level HostIDNode. To meet the persistence requirements of an XNS ID, this cross-reference must 341 
contain either: a) the IdentityID of an host identity defined in a different XNS addressing 342 
community, or b) a URN as defined in IETF RFC 2141 [5].. Like all cross-references, the IdentityID 343 
or URN is enclosed in parentheses. One common form of a URN that works well for a peer-to-peer 344 
XNS addressing community is a UUID, a 36- hex-character string generated according to a known 345 
specification so that for all practical purposes it is guaranteed to be globally unique (the probability 346 
of collision is infinitesimally small).  347 

Following are two examples of IdentityIDs in which the root HostIDNodes are cross-references 348 
expressed as URNs. The first is a UUID-based URN. The second is a URN system called Handle 349 
operated by the Corporation for National Research Initiatives (CNRI) [7]. 350 

:(urn:uuid:5a389ad2-22dd-11d1-aa77-002035b29092): 351 
:(urn:hdl:4263537/4090): 352 
 353 

Because they are native to XNS, OID-based IdentityIDs tend to be shorter. Examples: 354 

:4   (first-level host identity) 355 
:4.781   (second-level host registered with host identity :4) 356 
:4.781.23   (third-level host registered with host identity :4.781) 357 
 358 

Lines 3 and 4 allow any node in the host identity to be a cross-reference. For example, a cross-359 
reference containing a URN could be the top-level host identity node and OIDs could be used for 360 
lower-level nodes. Examples: 361 

:(urn:uuid:5a389ad2-22dd-11d1-aa77-002035b29092).781.23: 362 
:(urn:hdl:4263537/4090).781.23: 363 
 364 

To address any identity inside a host community, the host IdentityID is followed by a second colon 365 
followed by the hosted identity. The hosted identity is a path of one or more IdentityIDNodes 366 
delimited by dots. Note that lines 7 and 8 specify that the IdentityIDNode of a hosted identity can be 367 
any ID value that meets the XNS ID Normalization Rules. These rules are much looser than those for 368 
host identities, so while the ID of a hosted identity will typically be an integer, it may also be any 369 
other indexing value including the keys commonly used in SQL databases, LDAP directories, and 370 
other data stores. This avoids the need for identity documents to maintain the overhead of mapping 371 
XNS IDs to native data store keys. 372 

Examples of IdentityIDs for hosted identities: 373 

:4.781:560.73 374 
:4.781.23:hbrown44 375 
:(urn:uuid:5a389ad2-22dd-11d1-aa77-002035b29092).781.23:560.73 376 
:(urn:hdl:4263537/4090):hbrown44 377 

 378 

IdentityIDNodes, like HostIDNodes, may be expressed as cross-references. This allows, for example, 379 
an identity in one host community to be addressed by the IdentityID it is known by in another host 380 
community (provided the identity controller has given consent for this linkage). To separate it as an 381 
opaque indexing value, all cross-references are enclosed in parentheses. Examples: 382 



XNS Addressing Specification v1.1 26 March 2003  

© 2002-2003 XNSORG  Page 12  
 

:4.781:(:25.754:38056) 383 
:4.781.23:(:25.754:hbrown44) 384 
:(urn:uuid:5a389ad2-22dd-11d1-aa77-002035b29092):(:25.754:38056) 385 
:(urn:hdl:4263537/4090):(:(urn:hdl:559732/1246):hbrown44) 386 

5.2.4 Identity Data IDs 387 

[ 9] IdentityDataID ::= IdentityID DataID 388 
 389 
An IdentityDataID is simply an IdentityID concatenated with a DataID, explained in the productions 390 
below. Examples of IdentityDataIDs: 391 

:4.781:560.73;14.3 392 
:4.781.23:hbrown44;14.3 393 
:(urn:hdl:4263537/4090):hbrown44;email.home 394 

5.2.5 Data IDs 395 

[10] DataID ::= ';' DataIDNode [RelativeDataID] 396 
 397 
A DataID begins with a semicolon followed by at least one DataIDNode. This can be followed by 398 
the optional RelativeDataID of any lower-level element. Standing alone, a DataID is always relative 399 
to the root node of the current identity document. To make it absolute, it is combined with an 400 
IdentityID to form an IdentityDataID (above). Examples of DataIDs: 401 

;14 402 
;14.3 403 
;14.3.7 404 
;14.homePhone    (legal but not advised) 405 
;email.homePhone  (legal but not advised) 406 
 407 

Note that the last two examples use semantic characters as allowed by the XNS ID Normalization 408 
Rules. Although technically legal, this practice is strongly discouraged because XNS IDs, like all 409 
URNs, must continue to reference the same resource in spite of changing semantic relationships.  410 

5.2.6 Relative Data IDs 411 

[11] DataIDNode ::= ID | ('(' (IdentityDataID  | URN) ')') 412 
[12] RelativeDataID ::= *('.' DataIDNode) [',' Version] 413 
 414 

A RelativeDataID is any XNS ID that is relative to a node below the identity document root node. To 415 
syntactically distinguish them from XNS names, a RelativeDataID always begins with a dot. It can 416 
include any number of DataIDNodes, each delimited with a dot. Examples: 417 

.7 418 

.7.29 419 

.7.29.4 420 

.homePhone  (legal but not advised) 421 
 422 

Line 11 permits any form of a DataID to include a cross-reference at any data ID node. Examples of 423 
an IdentityDataID, a DataID, and a RelativeDataID that use cross-references: 424 

:4.781:560.73;(:732.41:28558;17).3 425 
;14.(:732.41:28558;17.8) 426 
.7.29.(:732.41:28558;17.8) 427 

5.2.7 Versions 428 

[13] Version ::= ('v' VersionNumber) | ('t' VersionDate) 429 
[14] VersionNumber ::= Non-negative integer 430 
[15] VersionDate ::= XML DateTime instance 431 
 432 
As explained in Object Versions, above, any XNS ID path to a data node (an IdentityDataID, 433 
DataID, or RelativeDataID) can include a version value to identify a specific version of the attribute 434 
associated with the data node. The version value is appended to the data ID path following a comma, 435 
and is prefixed with a “v” for integer format or “t” for XML dateTime format (see Object Versions, 436 
above). 437 

Examples of XNS IDs that include version values in both formats: 438 

:4.781:560.73;14.3,v3 439 
:4.781:560.73;14.3,v4 440 
;7.29,t2001-03-04T20:15:40Z 441 
;7.29,t2001-06-21T07:33:48Z 442 

5.2.8 XNS Names 443 

[16] XNSName ::= IdentityName | IdentityDataName | DataName | RelativeDataName 444 
 445 
As explained in the IDs, Names, and Addresses section above, an XNS name is a non-persistent 446 
identifier for an identity or identity attribute. It can be one of four types depending on whether it is 447 
the absolute name for the root node of an identity document (IdentityName), the absolute name for 448 
an element below the root node of an identity document (IdentityDataName), a name relative to the 449 
root node of an identity document (DataName), or a name relative to the current node of an identity 450 
document (RelativeDataName). 451 

5.2.9 Identity Names 452 

[17] IdentityName ::= (NamespaceSymbol | '//') IdentityNameNode  453 
*('/' IdentityNameNode) 454 
[18] NamespaceSymbol ::= '=' | '@' | '+' 455 
[19] IdentityNameNode ::= Name | ('(' IdentityAddress | URI ')') 456 
[20] Name ::= XML character string normalized according to the XNS Name 457 
Normalization Rules 458 
 459 

Because XNS identity names can be used as a human-friendly identity address—a consolidation of 460 
all other addressing attributes associated with an identity (phone number, email address, postal 461 
address, web address, instant messaging address, etc.)—the design goal is to make XNS name syntax 462 
as close to natural human language as possible. The result in line 17 is very similar to the Unix 463 
filename syntax widely used in URIs with four key differences: 464 

1. Three identity namespace prefix symbols are supported to indicate the three XNS-defined 465 
absolute namespaces (line 18). By comparison with DNS top-level domains (.com, .net, .org, 466 
.cc, .tv, etc.), these three identity namespace prefix symbols provide the shortest and simplest 467 
possible metadata necessary to establish the global context of an identity name. (See below.) 468 

2. Identity names can contain cross-references to other identities (line 19). This capability is 469 
very useful in federated identity management. The cross-reference can be expressed as either 470 
an IdentityAddress or a URI. (See examples below.) 471 

3. Namestrings can be any legal XML characters as defined by the W3C XML 1.0 Rec-472 
ommendation [2], i.e., they can use the full Unicode character set (see 473 
http://www.w3.org/TR/REC-xml#charsets). In addition, the design goal of the normalization 474 
rules for identity names is to permit maximum expressiveness while still meeting the 475 
minimum requirements for distinguishability of names—see the XNS Name Normalization 476 
Rules. This enables XNS identity names to be fully internationalized. 477 

4. Names for data objects can be versioned using the same syntax as XNS ID versioning. 478 

An IdentityName is a path that can begin with either: a) one of the three identity namespace prefix 479 
symbols (“=”, “@”, and “+”), or b) a double forward slash (“//”) representing the abstract XNS 480 
naming root. The three identity namespace prefix characters are simply shortcuts that expand into the 481 
full pathname following the Namespace Symbol Expansion rule, below. These three namespaces 482 
represent the three fundamental types of identity controllers in  the community rooted on //xns: 483 

 The Personal namespace (symbol “=”, which expands to “//xns/per/”) is reserved for 484 
names registered to represent individuals. These names do not have associated intellectual 485 
property rights.  486 

 The Organizational namespace (symbol “@”, which expands to “//xns/org/”) is reserved 487 
for names registered to represent any form of legal entity that is not an individual—sole 488 
proprietorships, partnerships, corporations, non-profits, governments, academic institutions, 489 
etc. Organizational names, also called business names, have associated intellectual property 490 
rights.  491 

 The General namespace (symbol “+”, which expands to “//xns/gen/”) is reserved for 492 
generic names that represent concepts or objects defined by the general public. In trademark 493 
law, generic names used in a generic context do not have associated intellectual property 494 
rights. XNSORG or its delegate acting as a public trustee registers generic names in the XNS 495 
general namespace.  496 

Note that in parsing, a namespace symbol is NOT considered a character in an XNS name value. The 497 
namespace symbol is expanded to its corresponding name path, and parsing continues with the 498 
subsequent XNS name value. Thus a namespace symbol character used as the literal first character of 499 
an XNS name must be escaped. See the XNS Name Normalization Rules, below. 500 

Following the identity namespace symbol or path is at least one XNS name string, which is any set 501 
of XML characters normalized according to the XNS Name Normalization Rules, below. This can be 502 
followed by any number of additional XNS namestrings, each delimited by forward slashes. 503 

Examples of XNS personal identity names using both namespace symbols and their expanded 504 
equivalents: 505 

=John 506 
//xns/per/John 507 
=John Smith 508 
//xns/per/John Smith 509 
=John Smith, Jr. 510 
//xns/per/John Smith, Jr. 511 
 512 

Examples of XNS organizational identity names: 513 

@Example 514 
@Example/Computers 515 
@Example/Computers/Internet 516 
@Smith & Jones 517 
@John Smith Inc. 518 
//xns/org/John Smith, Inc. 519 
 520 

Note that in the second and third examples above, the identity names are hierarchical: the identity 521 
@Example has registered the name “Computers” for another identity, and that identity has registered 522 
the name “Internet” for a third identity. Identity names can be hierarchical to any depth. 523 

Examples of XNS general identity names: 524 

+xns 525 
+plumber 526 
+Dominican Republic 527 
//xns/gen/Dominican Republic 528 
 529 

Examples of identity names using international character sets: 530 

=José Villegas, Jr. 531 
@A La François 532 
 533 

Examples of identity names using  IdentityAddresses as cross-references: 534 

@Example/(=John Smith) 535 
@Smith & Jones/(+garden rakes) 536 
=John Smith/(+email) 537 
 538 

Example of identity names using a URI: 539 

//(mailto:john.smith@example.com)/data/tel/Work 540 
 541 

5.2.10 Identity Addresses 542 

[21] IdentityAddress ::= (IdentityID ['!' IdentityName]) | IdentityName 543 
 544 
Line 19 in the EBNF allows a cross-reference to be not just another IdentityName, but an Iden-545 
tityAddress. An IdentityAddress is any combination of an IdentityID and an IdentityName that 546 
absolutely identifies an identity. If an IdentityID is present, then the IdentityName is optional and 547 
delimited by a bang sign (“!”) to indicate that it is only a human-readable comment—the Identity ID 548 
is always authoritative. This is useful for many contexts (e.g., web pages, software programs, 549 
reference manuals, etc.) where the persistence of an identity ID path is needed yet it is also desirable 550 
for it to be human readable without requiring resolution. 551 

If an IdentityID is not present, then an IdentityAddress must contain an IdentityName, which can 552 
then be resolved to the authoritative IdentityID. Examples of IdentityAddresses: 553 

:230.59:4.13.7421!=John Smith, Jr. 554 
:(urn:uuid:5a389ad2-22dd-11d1-aa77-002035b29092)!@Smith & Reilly 555 
:3.896324!+plumber 556 
 557 

Note that since an XNS identity controller may register multiple names for an identity, there may be 558 
more than one authoritative identity name to use with an identity address. The choice of identity 559 
name must be made by the address author. 560 

5.2.11 URI 561 

[22] URI ::= Uniform Resource Identifier as specified in IETF RFC 2396 [4] 562 

 563 

A URI according to RFC 2396 [4]. The full BNF is available in that document. Being able to use a 564 
URI as a cross-reference is one of the most powerful features of XNS Addressing, as it permits the 565 
XNS identity of any resource with an existing URI today to be located using that URI. 566 

5.2.12 Identity Data Names 567 

[23] IdentityDataName ::= IdentityName DataName 568 
 569 
As with IdentityDataIDs, an IdentityDataName is simply an IdentityName concatenated with a 570 
DataName, explained in the productions below. Examples of IdentityDataNames: 571 

=John Smith, Jr./Email/Home 572 
@Example/Computers/Internet/FTP 573 
+plumber/Hourly Rate 574 
 575 

Note that in the second example above, it is ambiguous whether any name after “@Example” (i.e., 576 
“Computers”, “Internet”, or “FTP”) is an identity name or a data name. Only by resolving the name 577 
to the underlying XNS ID can it be determined whether the target node is an identity node or a data 578 
node. 579 

5.2.13 Data Names 580 

[24] DataName ::= '/' RelativeDataName 581 
 582 
Standing alone, a DataName is always relative to the root node of the current identity document. 583 
Like a Unix filename that is relative to the root directory of the current drive, a DataName always 584 
begins with a single forward slash followed by a RelativeDataName. To make it absolute, a 585 
DataName is prefixed by with an IdentityName to form an IdentityDataName. Examples of 586 
DataNames: 587 

/Family/Father's side/Uncles/John 588 
/Uncles/John 589 
/John 590 

5.2.14 Relative Data Names 591 

[25] RelativeDataName ::= DataNameNode *('/' DataNameNode) ['/,' Version] 592 
[26] DataNameNode  ::= Name | ('(' (IdentityDataAddress | URI) ')') 593 
 594 
RelativeDataNames are just like relative path names in Unix with the exception of the richer XML 595 
character set and the ability to include cross-references and version metadata. RelativeDataNames do 596 
not have any leading delimiter and use forward slashes to delimit name nodes. Examples: 597 

Father's side/Uncles/John 598 
Uncles/John 599 
John 600 
 601 

To provide the same versioning capability as XNS IDs, the same versioning syntax can be appended 602 
to an XNS data name after a final forward slash: a comma, followed by “v” for an integer version 603 
value or “t” for an XML time instant. Examples: 604 

/Family/Father's side/Uncles/John/Phone/,v3 605 
John/Phone/,v4 606 
@Smith & Jones/Inventory/(+garden rakes)/,t2001-03-04T20:15:40Z 607 
=John Smith Jr./Phone/Work/,t2001-06-21T07:33:48Z 608 
 609 

Line 26 specifies that data names can also incorporate cross-references which themselves can be 610 
either IdentityDataAddresses (below) or URIs. Examples:  611 

@Yahoo/Computers/Internet/(@IBM/Computers/AS400) 612 
@Smith & Reilly/Tools/(+garden rakes/price) 613 
=John Smith/Friends/(=Mary Frank/Tel/Home) 614 
=John Smith/Friends/(mailto:mary.frank@example.com)/Tel/Home 615 
=John Smith/Friends/(http://www.maryfrank.com)/Tel/Home 616 
 617 

5.2.15 Identity Data Addresses 618 

[27] IdentityDataAddress ::= (IdentityDataID ['!' IdentityDataName]) | 619 
IdentityDataName 620 
 621 
Like cross references in identity name nodes (line 19), a cross-reference in a data name node needs to 622 
be able to include either an IdentityDataID or an IdentityDataName. Similar to an IdentityAddress 623 
(line 21), an IdentityDataAddress can be any combination of an IdentityDataID and an Identi-624 
tyDataName that absolutely identifies a data node within an identity. If an IdentityDataID is present, 625 
then the IdentityDataName is optional and the bang sign (“!”) indicates that it is only a human-626 
readable comment. If an IdentityDataID is not present, then an IdentityDataAddress must contain an 627 
IdentityDataName, which can be resolved to the authoritative IdentityDataID. Examples: 628 

:230.59:4.13.7421;14.2!=John Smith, Jr./Email/Work 629 
:(urn:hdl:4263537/4090):custACME;AEFF3CB.3956!@Acme/Eastern/Boats/ 630 
:3.896324:2499;77.98103!+plumber/(+flood repair)/Zip Code/98103 631 

5.2.16 Absolute Addresses 632 

[28] AbsoluteAddress ::= IdentityAddress | IdentityDataAddress 633 
 634 
Lastly, an AbsoluteAddress is a composite datatype allowing either an IdentityAddress or an Iden-635 
tityDataAddress. This datatype is useful for specifying an XNS address that must be absolute but can 636 
be either an XNS ID or XNS name and can resolve to either an identity node or a data node within an 637 
identity. 638 

6 XRI (XNS Resource Identifier) EBNF Definition 639 

To be useable on the Web, an XNS address must first be specified in URI format. This type of a URI 640 
is called an XRI (XNS Resource Identifier).  641 

XRIs also include the capability to invoke an XNS service associated with the target resource just as 642 
URIs using the HTTP schema can include query parameters following a question mark. XRIs use the 643 
same question mark syntax.  644 

When an XNS address is encoded as a URI according to IETF RFC 2396 [4], it MUST conform to 645 
the following EBNF definition. 646 

[01] XRI ::= URIScheme (ServiceCall | IdentityAddress | IdentityDataAddress) 647 
[02] URIScheme ::= HTTP | URN | XNS 648 
[03] HTTP ::= ['http://' | 'https://'] XNSResolverHostAddress '/xns:' 649 
[04] XNSResolverHostAddress ::= DNS or IP address of XNS resolver host 650 
[05] URN ::= 'urn:xns:' 651 
[06] XNS ::= 'xns:' 652 
[07] ServiceCall ::= '?' MessageAddress '(' [Argument] *(';' Argument) ')' 653 
[08] MessageAddress ::= IdentityDataAddress of XNS message definition 654 
[09] Argument ::= ArgName '=' ArgValue 655 
[10] ArgName ::= The argument name as defined by the message specification 656 
[11] ArgValue ::= The value of the argument as a string, with ';', ')', and '\' 657 
characters escaped with a '\' 658 
[12] IdentityAddress ::= as defined in XNS Address EBNF 659 
[13] IdentityDataAddress ::= as defined in XNS Address EBNF 660 
 661 
The three URI prefixes correspond to the three URI schemes [4] that will most commonly be used 662 
with XNS addresses: 663 

• The HTTP scheme is used to direct XNS address resolution requests to a resolver available 664 
at a DNS or IP address, i.e., any address recognized by the HTTP URI scheme.  665 

• The URN scheme is used to direct XNS address resolution requests to a URN resolver. 666 
NOTE: only XNS addresses that consist entirely of IdentityIDs or IdentityDataIDs qualify as 667 
URNs due to the persistence requirements of URNs [5]. 668 

• The XNS scheme is the native URI scheme for XNS, and presumes the URI parser 669 
understands XNS addressing. 670 

Note that in the HTTP and URN schemes, it is the native XNS URI scheme “xns:” that delimits the 671 
start of the XNS address string. In the HTTP scheme, these must be the first four characters 672 
following the forward slash that terminates the XNS resolver host address. 673 

7 XNS ID Normalization Rules 674 

The XNS Address EBNF establishes the strict syntax for the IDs used for host identities (they must 675 
be either OIDs or URNs). However the global rules for legal characters and normalization of XNS 676 
ID values at the identity or identity attribute (data) levels are intended to be looser to permit the use 677 
of a wide variety of conventional database keys, and to also allow identity controllers to establish 678 
their own stricter normalization rules for specific ID spaces. 679 

The question of what are the optimal ID normalization rules to impose on all XNS implementations 680 
is one on which XNSORG seeks community feedback through the forums available at www.xns.org. 681 
XNSORG expects to publish a formal EBNF definition for XNS ID normalization at a future date. 682 
Until then, the following top-level rules are normative: 683 

7.1 Legal XML Characters in IDs 684 

A normalized XNS ID value MUST NOT include any character defined as an illegal character by the 685 
W3C XML 1.0 Recommendation [2]. 686 

7.2 Unambiguous IDs 687 

A normalized XNS ID value MUST NOT include any character which causes ambiguity in parsing 688 
the ID value according to the EBNF definition of XNS addressing syntax. 689 

7.3 XNS Global Community ID 690 

All XNS ID resolvers SHOULD be able to internally resolve the following XNS Global Community 691 
ID: 692 

 693 

"::" - resolves to the following list of URIs: 694 
 http://resolver.xns.org/xns  695 
  https://resolver.xns.org/xns 696 
 697 

In addition, XNS ID resolvers SHOULD be able to internally resolve the following IDs: 698 

 699 
":1:" - resolves to the following list of URIs: 700 
  http://per.xns.org/xns  701 
  https://per.xns.org/xns 702 
 703 

":2:" - resolves to the following list of URIs: 704 
  http://org.xns.org/xns  705 
  https://org.xns.org/xns 706 
 707 

":3:" - resolves to the following list of URIs: 708 
  http://gen.xns.org/xns  709 
  https://gen.xns.org/xns  710 
 711 

8 XNS Name Normalization Rules 712 

Because it involves human semantics, internationalization, and the Unicode character set, XNS name 713 
normalization is a much more complex subject than ID normalization. Again the intention is to 714 
establish a baseline set of global rules for all implementations that can be further restricted within 715 
delegated namespaces. For the namespaces under its governance, the ultimate goal of XNSORG is to 716 
define name normalization rules that would identically normalize namestrings that a typical speaker 717 
of the relevant language would consider semantically equivalent. 718 

XNSORG invites community feedback on composing the XNS name normalization rules through the 719 
forums available at www.xns.org. XNSORG expects to publish a formal EBNF definition for XNS 720 
Name normalization. Until then, the following top-level rules are normative: 721 

8.1 Legal XML Characters in Names 722 

A normalized XNS Name value MUST NOT include any character defined as an illegal character by 723 
the W3C XML 1.0 Recommendation [2]. 724 

8.2 Unambiguous Names 725 

A normalized XNS Name value MUST NOT include any character which causes ambiguity in parsing 726 
the name value according to the EBNF definition of XNS addressing syntax. 727 

8.3 XML Letters and Digits 728 

A normalized XNS Name value MUST NOT include any character that is not classified as either a 729 
Letter or Digit according to Appendix B of the W3C XML 1.0 Recommendation [2]. 730 

8.4 Escape Character 731 

The ASCII character 092 decimal (backslash “\”) MUST be used to escape any character used in an 732 
XNS Name value which would not otherwise be allowed by the XNS name normalization rules, 733 
including this character itself. 734 

8.5 XNS Reserved Namespace 735 

The absolute namespace //xns/ is reserved and MUST be used only as specified by the XNS Public 736 
Trust Organization, which manages this namespace on behalf of the XNS community. 737 

This rule ensures that there is at least one globally interoperable namespace for addressing and cross-738 
referencing supported across all XNS implementations. 739 

8.6 Namespace Symbol Expansion 740 

In the XNS EBNF, the namespace symbol “=” MUST be expanded to the name path “//xns/per/”; 741 
the namespace symbol “@” MUST be expanded to the name path “//xns/org/”; the namespace 742 
symbol “+” MUST be expanded to the name path “//xns/gen/”. This expansion MUST be performed 743 
before applying EBNF parsing rules to the XNS name following the namespace symbol. 744 

This rule ensures that namespace symbols used in XNS identity names are interpreted correctly by 745 
XNS parsers. 746 

